John Guckenheimer & Philip Holmes Page 132–133, center manifold reduction

```
> restart:with(LinearAlgebra):
> # original system dot(u)=f1; dot(v)=f2
> f1:=v;
```

$$f1 := v \quad (1)$$

```
> f2:=-v+alp*u^2+bet*u*v;
```

$$f2 := alp\,u^2 + bet\,u\,v - v \quad (2)$$

```
> # fixed point
> fp:= solve({f1,f2},{u,v});
```

$$fp := \{u = 0, v = 0\} \quad (3)$$

```
> u0:=0:v0:=0:
> # Jacobian at the fixed point
> J:= Matrix(2,2,[[diff(f1,u),diff(f1,v)],[diff(f2,u),diff(f2,v)]])
;
```

$$J := \begin{bmatrix} 0 & 1 \\ 2\,alp\,u + bet\,v & bet\,u - 1 \end{bmatrix} \quad (4)$$

```
> J0:= simplify(subs(u=u0,v=v0,J));
```

$$J0 := \begin{bmatrix} 0 & 1 \\ 0 & -1 \end{bmatrix} \quad (5)$$

```
> ev:= Eigenvectors(J0);
```

$$ev := \begin{bmatrix} -1 \\ 0 \end{bmatrix}, \begin{bmatrix} -1 & 1 \\ 1 & 0 \end{bmatrix} \quad (6)$$

```
> lam1:= ev[1][2];lam2:=ev[1][1];
```

$$lam1 := 0$$

$$lam2 := -1 \quad (7)$$

```
> v1:= ev[2][1..2,2]; v2:= ev[2][1..2,1];
```

$$v1 := \begin{bmatrix} 1 \\ 0 \end{bmatrix}$$

$$v2 := \begin{bmatrix} -1 \\ 1 \end{bmatrix} \quad (8)$$

```
> T:= <v1|v2>; whattype(T);
```

$$T := \begin{bmatrix} 1 & -1 \\ 0 & 1 \end{bmatrix}$$

Matrix (9)

```
> TI:= MatrixInverse(T);
```

(10)

$$TI := \begin{bmatrix} 1 & 1 \\ 0 & 1 \end{bmatrix} \quad (10)$$

```
> TestT:= MatrixMatrixMultiply(TI,MatrixMatrixMultiply(J0,T));
```

$$TestT := \begin{bmatrix} 0 & 0 \\ 0 & -1 \end{bmatrix} \quad (11)$$

```
> # New variables (x,y)= TI * (u,v)
> X:= Vector(2,[u,v]);
```

$$X := \begin{bmatrix} u \\ v \end{bmatrix} \quad (12)$$

```
> Y:= Vector(2,[x,y]);
```

$$Y := \begin{bmatrix} x \\ y \end{bmatrix} \quad (13)$$

```
> X2:= MatrixVectorMultiply(T,Y);
```

$$X2 := \begin{bmatrix} x - y \\ y \end{bmatrix} \quad (14)$$

```
> F:= Vector(2,[f1,f2]);
```

$$F := \begin{bmatrix} v \\ alp\, u^2 + bet\, u\, v - v \end{bmatrix} \quad (15)$$

```
> G:= subs(u=X2[1],v=X2[2],MatrixVectorMultiply(TI,F));
```

$$G := \begin{bmatrix} alp\,(x-y)^2 + bet\,(x-y)\,y \\ alp\,(x-y)^2 + bet\,(x-y)\,y - y \end{bmatrix} \quad (16)$$

```
> g1:= G[1]; g2:= G[2];
```

$$g1 := alp\,(x-y)^2 + bet\,(x-y)\,y$$

$$g2 := alp\,(x-y)^2 + bet\,(x-y)\,y - y \quad (17)$$

```
> # dot(x)=g1; dot(y)=g2;
> TestJ0:= subs(x=0,y=0,Matrix(2,2,[[diff(g1,x),diff(g1,y)],[diff
  (g2,x),diff(g2,y)]]));
```

$$TestJ0 := \begin{bmatrix} 0 & 0 \\ 0 & -1 \end{bmatrix} \quad (18)$$

```
> # TestJ0=J0
> # assume the center manifold y=h(x)= a*x^2+b*x^3 + h.o.t.
> h:= a*x^2+b*x^3; dh:= diff(h,x);
```

$$h := b\,x^3 + a\,x^2$$

$$dh := 3\,b\,x^2 + 2\,a\,x \quad (19)$$

```
> N:= collect(expand(subs(y=h,dh*g1-g2)),x);
```

$$N := (3\,alp\,b^3 - 3\,b^3\,bet)\,x^8 + (8\,a\,alp\,b^2 - 8\,a\,b^2\,bet)\,x^7 + (7\,a^2\,alp\,b - 7\,a^2\,b\,bet - 7\,alp\,b^2 + 4\,b^2\,bet)\,x^6 + (2\,a^3\,alp - 2\,a^3\,bet - 12\,a\,alp\,b + 7\,a\,b\,bet)\,x^5 + (-5\,a^2\,alp + 3\,a^2\,bet + 5\,alp\,b - b\,bet)\,x^4 + (4\,a\,alp - a\,bet + b)\,x^3 + (a - alp)\,x^2 \qquad (20)$$

```
> N_x_2:= coeff(N,x,2);
```

$$N\_x\_2 := a - alp \qquad (21)$$

```
> N_x_3:= coeff(N,x,3);
```

$$N\_x\_3 := 4\,a\,alp - a\,bet + b \qquad (22)$$

```
> soln:= solve({N_x_2,N_x_3},{a,b})
```

$$soln := \{a = alp,\ b = -4\,alp^2 + alp\,bet\} \qquad (23)$$

```
> a:= rhs(soln[1]);
```

$$a := alp \qquad (24)$$

```
> b:= rhs(soln[2]);
```

$$b := -4\,alp^2 + alp\,bet \qquad (25)$$

```
> gg1:= collect(expand(subs(y=h,g1)),x);
```

$$gg1 := (16\,alp^5 - 24\,alp^4\,bet + 9\,alp^3\,bet^2 - alp^2\,bet^3)\,x^6 + (-8\,alp^4 + 10\,alp^3\,bet - 2\,alp^2\,bet^2)\,x^5 + (9\,alp^3 - 7\,alp^2\,bet + alp\,bet^2)\,x^4 + (-2\,alp^2 + alp\,bet)\,x^3 + alp\,x^2 \qquad (26)$$

```
> # the flow on the center manifold is dot(x)=gg1
```